

Gramática -GYH

`Programa → ':' 'DEC' ListaDeclaracoes ':' 'PROG' ListaComandos;`

`ListaDeclaracoes → Declaracao OutrasDeclaracoes;`

`OutrasDeclaracoes → ListaDeclaracoes | ε;`

`Declaracao → VARIABEL ':' TipoVar;`

`TipoVar → 'INT' | 'REAL';`

`ExpressaoAritmetica → TermoAritmetico SentencaAritmetica;`

`SentencaAritmetica → '+' TermoAritmetico SentencaAritmetica | '-' TermoAritmetico SentencaAritmetica | ε;`

`TermoAritmetico → FatorAritmetico ProposicaoAritmetica;`

`ProposicaoAritmetica → '*' FatorAritmetico ProposicaoAritmetica | '/' FatorAritmetico ProposicaoAritmetica | ε;`

`FatorAritmetico → NUMINT | NUMREAL | VARIABEL | '(' ExpressaoAritmetica ')'`

`ExpressaoRelacional → TermoRelacional SentencaRelacional;`

`SentencaRelacional → OperadorBooleano TermoRelacional SentencaRelacional | ε;`

`TermoRelacional → ExpressaoAritmetica OP_REL ExpressaoAritmetica | '(' ExpressaoRelacional ')';`

`# Vai ter um operador relacional antes de fechar o parenteses, ao contrário da sentença aritmética`

`OperadorBooleano → 'E' | 'OU';`

`ListaComandos → Comando OutrosComandos;`

`OutrosComandos → ListaComandos | ε;`

`Comando → ComandoAtribuicao | ComandoEntrada | ComandoSaida | ComandoCondicao | ComandoRepeticao | SubAlgoritmo;`

`ComandoAtribuicao → VARIABEL ':= ' ExpressaoAritmetica;`

`ComandoEntrada → 'LER' VARIABEL;`

ComandoSaida → 'IMPRIMIR' TipoSaida;

TipoSaida → VARIAVEL | CADEIA;

ComandoCondicao → 'SE' ExpressaoRelacional 'ENTAO' Comando ContraCondicao;

ContraCondicao → 'SENAO' Comando | ε;

ComandoRepeticao → 'ENQTO' ExpressaoRelacional Comando;

SubAlgoritmo → 'INI' ListaComandos 'FIM';

Lexemas x Tokens

Padrão	Tipo de Lexema	Sigla
DEC, PROG, INT, REAL, LER, IMPRIMIR, SE, SENAO, ENTAO, ENQTO, INI, FIM	Palavras-chave	PC... (PCDec, PCProg, PCInt, PCReal, PCLer, PCImprimir, PCSe, PCSenao, PCEntao, PCEnqto, PCIni, PCFim)
*, /, +, -	Operadores Aritméticos	OpArit (OpAritMult, OpAritDiv, OpAritSoma, OpAritSub)
<, <=, >, >=, ==, !=	Operadores Relacionais	OpRel (OpRelMenor, OpRelMenorIgual, OpRelMaior, OpRelMaiorIgual, OpRelIgual, OpRelDif)
E, OU	Operadores booleanos	OpBoolE, OpBoolOu
:	Delimitador	Delim
:=	Atribuição	Atrib
(,)	Parêntesis	AbrePar, FechaPar,
Sequências de letras ou números que começam com letra minúscula	Variável	Var
Sequências de dígitos (sem ponto)	Número Inteiro	NumInt
Sequências de dígitos (com ponto)	Número Real	NumReal

Sequências de caracteres envolta por aspas	Cadeia de caracteres	Cadeia
--	----------------------	--------